Define basis functions using Lagrange interpolation. It suffices to compute on the interval (0,h), since other intervals can be transformed to (0,h) by a change of variable.

```
> basisfunctions := matrix(3,1,
                           [ (x-h/2)*(x-h) / (0-h/2) / (0-h),
                             (x-0)*(x-h) / (h/2-0) / (h/2-h),
                             (x-0)*(x-h/2) / (h-0) / (h-h/2)  ]);
```

$$basisfunctions := \begin{bmatrix} \frac{2\left(x-\frac{1}{2}h\right)(x-h)}{h^2} \\ -\frac{4x(x-h)}{h^2} \\ \frac{2x\left(x-\frac{1}{2}h\right)}{h^2} \end{bmatrix} \tag{1}$$

Compute the stiffness matrix:

```
> A := matrix(3,3):
  for i from 1 to 3 do
    for j from 1 to 3 do
      A [i,j] := simplify(int(
                 diff(basisfunctions[i,1],x)
                 *diff(basisfunctions[j,1],x),
                 x=0..h )):
    end do:
  end do:
  evalm(A);
```

$$\begin{bmatrix} \frac{7}{3h} & -\frac{8}{3h} & \frac{1}{3h} \\ -\frac{8}{3h} & \frac{16}{3h} & -\frac{8}{3h} \\ \frac{1}{3h} & -\frac{8}{3h} & \frac{7}{3h} \end{bmatrix} \tag{2}$$

Compute the mass matrix:

```
> M := matrix(3,3):
  for i from 1 to 3 do
    for j from 1 to 3 do
      M [i,j] := simplify(int(
                 basisfunctions[i,1]
                 * basisfunctions[j,1] ,
                 x=0..h )):
    end do:
```

```
end do:
evalm(M);
```

$$\begin{bmatrix} \frac{2}{15}h & \frac{1}{15}h & -\frac{1}{30}h \\ \frac{1}{15}h & \frac{8}{15}h & \frac{1}{15}h \\ -\frac{1}{30}h & \frac{1}{15}h & \frac{2}{15}h \end{bmatrix} \tag{3}$$

>